
1 Overview

This document lists all the changes needed to correct problems in the TTCN implementation of test case 7.1.3.5, which is part of the LTE test suite. 

The test case can be demonstrated to run with one LTE UE (see section 6). Execution log files are provided as evidence. 

3 Verification Test Summary

Test Case:
tc_7_1_3_5
Test Group:
RRC
ATS Version:
iwd-EUTRA-TVB2009-03_D09wk47
System Simulator used:
Anite Conformance Toolset
UE used:
LG Electronics LEO3 UE.

Verification Status:
PASS
4 Corrections required for test case 7.1.3.5
4.1 Introduction

This section describes the changes required to make test case 7.1.3.5 run correctly with a LTE UE. The TTCN used to pass this test case belongs to iwd-EUTRA-TVB2009-03_D09wk07 release.

4.2 Change 1
	Testcase name
	f_TC_7_1_3_5_EUTRA

	Reason for change
	During this test case DL Timing Advance can be sent by the SS and in such a case UE will send the Harq Ack for this DL Timing Advance. As Harq Ack/Nack events are checked in this test case, so this test case may fail if DL timing advance is sent by the SS.

	Summary of change
	f_EUTRA_StartDefULGrantTransmission is replaced by f_EUTRA_StartULGrantTransmissionOnSR_Reception_No_PUCCH_Sync

	Source of change
	File: MAC_713.ttcn

	MCC 160
	Accepted


Before:
MAC_713.ttcn : 

	function f_TC_7_1_3_5_EUTRA ( ) runs on EUTRA_PTC

  {
    …………………..
    …………………..
  // to configure SS to start report reception of HARQ ACK or NACK

    f_SS_ConfigUL_HARQ_IndMode (eutra_Cell1,tsc_L1Mac_IndicationMode_Enable );

    // Preconfig SS for default UL grant allocation on SR reception.

    f_EUTRA_StartDefULGrantTransmission(eutra_Cell1,cs_TimingInfo_Now);
    ///////////////////////////////////////////////////////////////////////////////////

    //      Test Body

    //////////////////////////////////////////////////////////////////////////////////
    …………………..
    …………………..
 }



After:
MAC_713.ttcn : 

	function f_TC_7_1_3_5_EUTRA ( ) runs on EUTRA_PTC

  {
    …………………..
    …………………..

    f_SS_ConfigUL_HARQ_IndMode (eutra_Cell1,tsc_L1Mac_IndicationMode_Enable );

    // Preconfig SS for default UL grant allocation on SR reception.

    f_EUTRA_StartULGrantTransmissionOnSR_Reception_No_PUCCH_Sync(eutra_Cell1, cs_TimingInfo_Now, cs_DciInfo_CcchDcchDtchUL_Def); 

    ///////////////////////////////////////////////////////////////////////////////////

    //      Test Body

    //////////////////////////////////////////////////////////////////////////////////

    f_EUTRA_TestBody_Set(true);
    …………………..
    …………………..
 }




4.3 Change 2

	Testcase name
	f_TC_7_1_3_5_EUTRA

	Reason for change
	1) At test step 18 and 20, there can be a race condition between HARQ ACK for MSG4 and RRC connection Request, as these are only 5 ms apart and RRC connection request goes through 2 ports, (SYS_SRB and SRB)
2) At step 21, SS waits for RRC Complete message. As RLC is in AM mode, so it sends the RLC ACK automatically and UE will send the HARQ ACK for this RLC ack. TTCN should handle the HARQ ACK for this RLC Ack.
3) Similar  to case 1 there can be race condition between RRC Complete and HARQ ACK received for RLC_ACK as they can be only 5 subframe apart

	Summary of change
	1) Race condition handled for HARQ ACK and RRC connection Request
2) After step 21, HARQ ACK for RLC ACK sent by the RLC, is handled
3) Race condition handled for HARQ ACK and RRC Complete is handled

	Source of change
	File: MAC_713.ttcn

	MCC 160
	Accepted in principle, implemented as shown below


Before:

	function f_TC_7_1_3_5_EUTRA ( ) runs on EUTRA_PTC

  {
    …………………..
    …………………..
    var RRC_TransactionIdentifier v_RRC_TI := tsc_RRC_TI_Def;

    var integer j :=0, k :=0;

    var EUTRA_FDD_TDD_Mode_Type  v_EUTRA_FDD_TDD_Mode;

    ///////////////////////////////////////////////////////////////////////////////////

    //      PREAMBLE

    //////////////////////////////////////////////////////////////////////////////////
    …………………..
    …………………..
    // step 17 automatically taken care of

    // step 18

    t_Watchdog.start;

    SRB.receive ( car_SRB0_RrcPdu_IND ( eutra_Cell1, cr_RRCConnectionRequest ( ? ) ) );

    f_EUTRA_PreliminaryPass(__FILE__, __LINE__, "");

    t_Watchdog.stop;

    // step 19 automatically taken care of

    // step 20

    t_Watchdog.start;

    SYSIND.receive ( car_UL_HARQ_IND (eutra_Cell1, ?, ack )); // step 20

    f_EUTRA_PreliminaryPass(__FILE__, __LINE__, "");

    t_Watchdog.stop;
    // step 21

    t_Watchdog.start;

    SRB.receive ( car_SRB1_RrcNasPdu_IND ( eutra_Cell1, cr_508_RRCConnectionSetupComplete ( tsc_RRC_TI_Def, ? ),? ) );

    f_EUTRA_PreliminaryPass(__FILE__, __LINE__, "");

    t_Watchdog.stop;
    f_EUTRA_TestBody_Set(false);
    …………………..
    …………………..
 }



After:

	function f_TC_7_1_3_5_EUTRA ( ) runs on EUTRA_PTC

  {
    …………………..
    …………………..

    var template (value) RachProcedureConfig_Type v_RachProcedureConfig;

    var boolean NotRRCComplete := true;
    var boolean NotHarqInd := true;

    var boolean NotRRCRequest:= true;
    ///////////////////////////////////////////////////////////////////////////////////

    //      PREAMBLE

    //////////////////////////////////////////////////////////////////////////////////
    …………………..
    …………………..

    // step 17 automatically taken care of

    // step 18

    t_Watchdog.start;

    alt {

        [NotRRCRequest]SRB.receive ( car_SRB0_RrcPdu_IND ( eutra_Cell1, cr_RRCConnectionRequest ( ? ) ) )

        {

            f_EUTRA_PreliminaryPass(__FILE__, __LINE__, "");

            if (NotHarqInd)

            {

                NotRRCRequest := false;

                repeat;

            }

            t_Watchdog.stop;

        }

    // step 19 automatically taken care of

    // step 20

        [NotHarqInd]SYSIND.receive ( car_UL_HARQ_IND (eutra_Cell1, ?, ack )) // step 20

        {

            f_EUTRA_PreliminaryPass(__FILE__, __LINE__, "");

            if (NotRRCRequest)

            {

                NotHarqInd := false;

                repeat;

            }

            t_Watchdog.stop;

        }

    }

    NotHarqInd := true;

    // step 21

    t_Watchdog.start;

    alt {

        [NotRRCComplete]SRB.receive ( car_SRB1_RrcNasPdu_IND ( eutra_Cell1, cr_508_RRCConnectionSetupComplete ( tsc_RRC_TI_Def, ? ),? ) )

        {

            f_EUTRA_PreliminaryPass(__FILE__, __LINE__, "");

            if (NotHarqInd)

            {

                NotRRCComplete := false;

                repeat;

            }

            t_Watchdog.stop;

        }

        [NotHarqInd]SYSIND.receive ( car_UL_HARQ_IND (eutra_Cell1, ?, ack )) 

        {

            if (NotRRCComplete)

            {

                NotHarqInd := false;

                repeat;

            }

            t_Watchdog.stop;

        }

    }
    f_EUTRA_TestBody_Set(false);
    …………………..
    …………………..
 }


	


MCC change
	function f_TC_7_1_3_5_EUTRA ( ) runs on EUTRA_PTC

  { /* L2/MAC/7.1

       7.1.3.5 : Correct HARQ process handling [CCCH]

    */

    var template (value) EUTRA_FDD_TDD_CellInfo_Type  v_EUTRA_FDD_TDD_Info;

    var octetstring v_EncodedRRCConnectionSetup;

    var template (value) RachProcedureConfig_Type v_RachProcedureConfig;

    // variables for handling racing problem @sic R5s090132 sic@

    var template (omit) SRB_COMMON_IND v_SrbIndByRef:=omit;    // not used (needed when other information of the RRC message needs to be access)

    var template (omit) SYSTEM_IND v_SysIndByRef:=omit;        // to store SYSTEM_IND if it takes over the RRC message

    var SR_PUCCH_ResourceIndex_Type v_Sr_PUCCH_ResourceIndex;

    var boolean v_RRCConReq := false, v_HARQ_Ind := false, v_HARQ_Ind2 := false, v_RRCConSetupCmpl := false;
    var SRB_COMMON_IND v_SRB_COMMON_IND;

    var SYSTEM_IND v_SYSTEM_IND;

    var SubFrameTiming_Type v_Timing_SysInd,v_Timing_SysInd2, v_Timing_SrbIndReq, v_Timing_SrbIndCmpl;
    timer t_Watchdog:=2.0; // General purpose wait timer set as T300 value

    ///////////////////////////////////////////////////////////////////////////////////

    //      PREAMBLE

    //////////////////////////////////////////////////////////////////////////////////

    f_EUTRA_Init (c1);

    v_EUTRA_FDD_TDD_Info:=f_EUTRA_CellInfo_GetFDD_TDD_Info (eutra_Cell1);

    v_Sr_PUCCH_ResourceIndex := f_EUTRA_CellInfo_GetSR_PUCCH_ResourceIndex ( eutra_Cell1 );

    // Modify SIB's for test case Specific values.

    f_EUTRA_CellInfo_SetPreambleTransMax(eutra_Cell1,n8 );

    f_EUTRA_CellInfo_SetSysInfo_T300 (eutra_Cell1, ms2000);

    f_EUTRA_CellInfo_SetMAC_ContResTimer (eutra_Cell1, sf64);

    f_EUTRA_CellConfig_Def (eutra_Cell1);

    f_EUTRA_Preamble ( eutra_Cell1 , STATE2_IDLEUPDATE);

    // creates cell and performs registration

    v_EncodedRRCConnectionSetup := f_RRC_DL_CCCH_Encvalue (cs_508_RRCConnectionSetup(tsc_RRC_TI_Def,

                                                                                     v_EUTRA_FDD_TDD_Info.pucch_Config,v_Sr_PUCCH_ResourceIndex));

    v_RachProcedureConfig := cs_RachProcedureConfig_7135 ( v_EncodedRRCConnectionSetup, f_EUTRA_CellInfo_GetRAR_TA(eutra_Cell1) );

                                                                                      /*

                                                                                      cs_RachProcedureConfig_7135 has four entries

                                                                                         step 3 CRC error

                                                                                         Step 9 not matching ContResolution ID

                                                                                         Step 19 CRC error

                                                                                         Step 19 success

                                                                                      */

    f_EUTRA_SS_ConfigRachProcedure(eutra_Cell1, omit, v_RachProcedureConfig );

    // Preconfigures SS for RACH response for RAR and contention resolution

    // Configure SS to enable report of PRACH preamble reception

    f_EUTRA_SS_ConfigRachPreambleIndMode (eutra_Cell1, tsc_L1Mac_IndicationMode_Enable);

    // to configure SS to start report reception of HARQ ACK or NACK

    f_SS_ConfigUL_HARQ_IndMode (eutra_Cell1, tsc_L1Mac_IndicationMode_Enable);

    // Preconfig SS for default UL grant allocation on SR reception.

    f_EUTRA_StartULGrantTransmissionOnSR_Reception_No_PUCCH_Sync(eutra_Cell1,cs_TimingInfo_Now, cs_DciInfo_CcchDcchDtchUL_Def);

    ///////////////////////////////////////////////////////////////////////////////////

    //      Test Body

    //////////////////////////////////////////////////////////////////////////////////

    f_EUTRA_TestBody_Set(true);

    // Test body starts

    // step 1

    f_EUTRA_UE_Page_Def ( eutra_Cell1 );

    // step 2

    t_Watchdog.start;

    SYSIND.receive ( car_PRACH_Preamble_IND_Any (eutra_Cell1, ?));

    f_EUTRA_PreliminaryPass(__FILE__, __LINE__, "");

    t_Watchdog.stop;

    // step 3 automatically taken care of

    // step 4 and 5

    t_Watchdog.start;

    alt {

      []SYSIND.receive ( car_UL_HARQ_IND (eutra_Cell1, ?, ? )) // step 4

        {

          t_Watchdog.stop;

          f_EUTRA_SetVerdictFailOrInconc(__FILE__, __LINE__, "UE sending a ACK/NACK");

        }

      []SYSIND.receive ( car_PRACH_Preamble_IND_Any (eutra_Cell1,?)) // step 5

        {

          f_EUTRA_PreliminaryPass(__FILE__, __LINE__, "");

          t_Watchdog.stop;

        }

    };

    // step 6 automatically taken care of

    /*

    // step 7 and 8

    t_Watchdog.start;

    alt {

      []SYSIND.receive ( car_UL_HARQ_IND (eutra_Cell1, ?, ack )) // step 7

        {

          t_Watchdog.stop;

          f_EUTRA_SetVerdictFailOrInconc(__FILE__, __LINE__, "UE sending a ACK");

        }

      [] SRB.receive ( car_SRB0_RrcPdu_IND ( eutra_Cell1, cr_RRCConnectionRequest ( ? ) ) ) // step 8

        {

          f_EUTRA_PreliminaryPass(__FILE__, __LINE__, "");

          t_Watchdog.stop;

        }

    };

    // step 9 automatically taken care of

    // step 10 and 11

    t_Watchdog.start;

    alt {

      []SYSIND.receive ( car_UL_HARQ_IND (eutra_Cell1, ?, nack )) // step 10

        {

          t_Watchdog.stop;

          f_EUTRA_SetVerdictFailOrInconc(__FILE__, __LINE__, "UE sending a NACK");

        }

      []SYSIND.receive ( car_PRACH_Preamble_IND_Any (eutra_Cell1,?)) // step 11

        {

          f_EUTRA_PreliminaryPass(__FILE__, __LINE__, "");

          t_Watchdog.stop;

        }

    };

    */

    //step 8 and possibly step 11.

    // step 7 and step 10 to match in default handler.

    //step 9 automatically taken care off

    f_EUTRA_RacingCond_AwaitRrcMessage(car_SRB0_RrcPdu_IND ( eutra_Cell1, cr_RRCConnectionRequest ( ? ) ),

                                       v_SrbIndByRef, v_SysIndByRef );

    f_EUTRA_PreliminaryPass(__FILE__, __LINE__, "");

    // If step 11 not received then receive else to match

    f_EUTRA_RacingCond_MatchOrReceiveSysInd(v_SysIndByRef,

                                            car_PRACH_Preamble_IND_Any (eutra_Cell1, ?));

    // Step 12 automatically taken care of

    // Note Racing should not be between steps 13 and 16 as UE has to wait for mac-ContentionResolutionTimer  which is 48 ms

    //before initiating the next PRACH procedure.

    // step 13

    t_Watchdog.start;

    SRB.receive ( car_SRB0_RrcPdu_IND ( eutra_Cell1, cr_RRCConnectionRequest ( ? ) ) ); // step 13

    f_EUTRA_PreliminaryPass(__FILE__, __LINE__, "");

    t_Watchdog.stop;

    // step 14 Automatically taken care of

    // step 15 and 16.

    t_Watchdog.start;

    alt {

      []SYSIND.receive ( car_UL_HARQ_IND (eutra_Cell1, ?, nack )) // step 15

        {

          t_Watchdog.stop;

          f_EUTRA_SetVerdictFailOrInconc(__FILE__, __LINE__, "UE sending a NACK");

        }

      []SYSIND.receive ( car_PRACH_Preamble_IND_Any (eutra_Cell1,?)) // step 16

        {

          f_EUTRA_PreliminaryPass(__FILE__, __LINE__, "");

          t_Watchdog.stop;

        }

    };

    // step 17 automatically taken care of

    t_Watchdog.start;

    alt {

      // step 18

      // step 19 automatically taken care of

      [not v_RRCConReq] SRB.receive ( car_SRB0_RrcPdu_IND ( eutra_Cell1, cr_RRCConnectionRequest ( ? ) ) ) -> value v_SRB_COMMON_IND

        {

          v_Timing_SrbIndReq := v_SRB_COMMON_IND.Common.TimingInfo.SubFrame; // timestamp when RRCConnectReq has been received

          v_RRCConReq := true;

          if (not (v_HARQ_Ind and v_RRCConSetupCmpl and v_HARQ_Ind2)) {repeat}

          else { t_Watchdog.stop; }

        }

      // step 20

      [not v_HARQ_Ind] SYSIND.receive ( car_UL_HARQ_IND (eutra_Cell1, ?, ack )) -> value v_SYSTEM_IND

        {

          f_EUTRA_PreliminaryPass(__FILE__, __LINE__, "");

          v_HARQ_Ind := true;

          v_Timing_SysInd := v_SYSTEM_IND.Common.TimingInfo.SubFrame;     // timestamp when SYSTEM_IND has been received

          if (not (v_RRCConReq and v_RRCConSetupCmpl and v_HARQ_Ind2)) {repeat}

          else { t_Watchdog.stop; }

        }

      // step 21

      [not v_RRCConSetupCmpl] SRB.receive ( car_SRB1_RrcNasPdu_IND ( eutra_Cell1, cr_508_RRCConnectionSetupComplete ( tsc_RRC_TI_Def, ? ), ? ) )-> value v_SRB_COMMON_IND

        {

          v_Timing_SrbIndCmpl := v_SRB_COMMON_IND.Common.TimingInfo.SubFrame; // timestamp when RRCConnectSetup complete has been received

          v_RRCConSetupCmpl := true;

          if (not ( v_RRCConReq and v_HARQ_Ind and v_HARQ_Ind2)) {repeat}// RRC connection shall be received before RRC connection setup complet;



                                                                 // checked to guarantee run time errors if any un initialised timing info check in if statement below;

          else { t_Watchdog.stop; }

        }


[not v_HARQ_Ind2] SYSIND.receive ( car_UL_HARQ_IND (eutra_Cell1, ?, ack )) -> value v_SYSTEM_IND


// HARQ for RLC Status PDU for RRC Connection Setup Complete

        {

          v_HARQ_Ind2 := true;

          v_Timing_SysInd2 := v_SYSTEM_IND.Common.TimingInfo.SubFrame;     // timestamp when SYSTEM_IND has been received

          if (not ( v_RRCConReq and v_HARQ_Ind and v_RRCConSetupCmpl )) {repeat} // note HARQ for Setup shall come before HARQ for HARQ for Status PDU

          else { t_Watchdog.stop; }

        }

    };

    if ( (f_EUTRA_SubFrameTimingDuration(v_Timing_SysInd, v_Timing_SrbIndReq )> 5120) or

         (f_EUTRA_SubFrameTimingDuration(v_Timing_SrbIndCmpl, v_Timing_SysInd )> 5120) or



 (f_EUTRA_SubFrameTimingDuration(v_Timing_SrbIndCmpl, v_Timing_SrbIndReq )> 5120) or

         (f_EUTRA_SubFrameTimingDuration(v_Timing_SysInd2, v_Timing_SrbIndCmpl )> 5120) or



 (f_EUTRA_SubFrameTimingDuration(v_Timing_SysInd2, v_Timing_SysInd )> 5120) or



 (f_EUTRA_SubFrameTimingDuration(v_Timing_SysInd2, v_Timing_SrbIndReq )> 5120))



 {




 f_EUTRA_SetVerdictInconc(__FILE__, __LINE__, "got unexpected message order"); 




 }
    // to configure SS to stop report reception of HARQ ACK or NACK

    f_SS_ConfigUL_HARQ_IndMode (eutra_Cell1,tsc_L1Mac_IndicationMode_Disable );

    //Send Service Reject with cause Congestion

    SRB.send ( cas_SRB1_NasPdu_REQ ( eutra_Cell1,

                                     cs_TimingInfo_Now,

                                     cs_NAS_Request ( tsc_SHT_NoSecurityProtection,

                                                      cs_508_SERVICE_REJECT ( tsc_Cause_Congestion ) ) ) );

    f_EUTRA_TestBody_Set(false);

    //test body ends

    ///////////////////////////////////////////////////////////////////////////////////

    //      Post Amble

    //////////////////////////////////////////////////////////////////////////////////

    // Configure SS to disable report of  PRACH preamble reception

    f_EUTRA_SS_ConfigRachPreambleIndMode (eutra_Cell1 ,tsc_L1Mac_IndicationMode_Disable);

    // Preconfigures SS for RACH response for default RAR and contention resolution

    f_EUTRA_SS_ConfigRachProcedure (eutra_Cell1, omit, cs_RachProcedureConfig_Def(omit, f_EUTRA_CellInfo_GetRAR_TA(eutra_Cell1)));

    //Switch off UE and expect a DETACH REQUEST message

    f_EUTRA_Postamble ( eutra_Cell1, RRC_CONNECTED, ATTACHED );

  };


4.4 Change 3

	Testcase name
	f_TC_7_1_3_5_EUTRA

	Reason for change
	During Postamble, SS sends service reject to the UE, As HARQ ACK Events are enabled, so the HARQ ACK for the service reject can also reach to the TTCN before the disable of the HARQ Events. So HARQ ACK events should be disabled before sending the Service Reject

	Summary of change
	f_SS_ConfigUL_HARQ_IndMode is moved before sending the service reject

	Source of change
	File: MAC_713.ttcn

	MCC 160
	Accepted


Before:

	function f_TC_7_1_3_5_EUTRA ( ) runs on EUTRA_PTC

  {
    …………………..
    …………………..
    //test body ends

    ///////////////////////////////////////////////////////////////////////////////////

    //      Post Amble

    //////////////////////////////////////////////////////////////////////////////////

    //Send Service Reject with cause Congestion

    SRB.send ( cas_SRB1_NasPdu_REQ ( eutra_Cell1,

                                     cs_TimingInfo_Now,

                                     cs_NAS_Request ( tsc_SHT_NoSecurityProtection,

                                                      cs_508_SERVICE_REJECT ( tsc_Cause_Congestion ) ) ) );

    // to configure SS to stop report reception of HARQ ACK or NACK

    f_SS_ConfigUL_HARQ_IndMode (eutra_Cell1,tsc_L1Mac_IndicationMode_Disable );
    // Configure SS to disable report of  PRACH preamble reception

    f_EUTRA_SS_ConfigRachPreambleIndMode ( eutra_Cell1 ,tsc_L1Mac_IndicationMode_Disable);

    // Preconfigures SS for RACH response for default RAR and contention resolution

    f_EUTRA_SS_ConfigRachProcedure (eutra_Cell1,omit, cs_RachProcedureConfig_Def(omit,f_EUTRA_CellInfo_GetRAR_TA(eutra_Cell1)));
    …………………..
    …………………..
 }



After:

	function f_TC_7_1_3_5_EUTRA ( ) runs on EUTRA_PTC

  {
    …………………..
    …………………..

   //test body ends

    ///////////////////////////////////////////////////////////////////////////////////

    //      Post Amble

    //////////////////////////////////////////////////////////////////////////////////

    // to configure SS to stop report reception of HARQ ACK or NACK

    f_SS_ConfigUL_HARQ_IndMode (eutra_Cell1,tsc_L1Mac_IndicationMode_Disable );//ANITE_LTE_204(b) End
    //Send Service Reject with cause Congestion

    SRB.send ( cas_SRB1_NasPdu_REQ ( eutra_Cell1,

                                     cs_TimingInfo_Now,

                                     cs_NAS_Request ( tsc_SHT_NoSecurityProtection,

                                                      cs_508_SERVICE_REJECT ( tsc_Cause_Congestion ) ) ) );

    // Configure SS to disable report of  PRACH preamble reception

    f_EUTRA_SS_ConfigRachPreambleIndMode ( eutra_Cell1 ,tsc_L1Mac_IndicationMode_Disable);

    // Preconfigures SS for RACH response for default RAR and contention resolution

    f_EUTRA_SS_ConfigRachProcedure (eutra_Cell1,omit, cs_RachProcedureConfig_Def(omit,f_EUTRA_CellInfo_GetRAR_TA(eutra_Cell1)));
    …………………..
    …………………..
 }



5 Execution Log Files

5.1 LG Electronics LEO3 UE

The LG electronics LEO3 UE passed this test case on the Anite Conformance Toolset solution with Snow 3G integrity algorithm. The documentation below is enclosed as evidence of the successful test case run [1]:

· Test Case Execution log file tc_7_1_3_5_LG-log.html:

In the log file (in html format) the complete test case execution can be seen. All message contents are fully decoded and can be verified. Preliminary verdicts and the final test case verdict can be seen in the log file.

6 References

	[1]
	R5s100075: This archive comprises text format execution log file.

